Mi-101 **Mathematica** 2004-2005

Correction Partiel - sujet 1

Exercice 1 : cours

■ Listes

1) En utilisant *Range*, donner une commandes permettant d'obtenir chacune des listes suivantes (on expliquera la propriété justifiant le résultat).

```
Log[Range[7]]
```

```
{0, Log[2], Log[3], Log[4], Log[5], Log[6], Log[7]}
```

```
Sqrt[Range[12, -2, -1]]
```

$\{2\sqrt{3}, \sqrt{11}, \sqrt{10}, 3, 2\sqrt{2}, \sqrt{7}, \sqrt{6}, \sqrt{5}, 2, \sqrt{3}, \sqrt{2}, 1, 0, i, i\sqrt{2}\}$

```
x * Range[5] + y * Range[2, 6]
```

```
{x + 2 y, 2 x + 3 y, 3 x + 4 y, 4 x + 5 y, 5 x + 6 y}
```

2) En utilisant *Table*, donner une commandes permettant d'obtenir chacune des listes suivantes.

```
Table[i^toto, {i, -3, 3}]
```

$\{(-3)^{toto}, (-2)^{toto}, (-1)^{toto}, 0^{toto}, 1, 2^{toto}, 3^{toto}\}$

```
Table[a^i - b^i, {i, 3, 7}]
```

$\{a^3 - b^3, a^4 - b^4, a^5 - b^5, a^6 - b^6, a^7 - b^7\}$

```
Table[a / i - b^j, {i, 5, 7}, {j, 1, 2}]
```

$\left\{\left\{\frac{a}{5} - b, \frac{a}{5} - b^2\right\}, \left\{\frac{a}{6} - b, \frac{a}{6} - b^2\right\}, \left\{\frac{a}{7} - b, \frac{a}{7} - b^2\right\}\right\}$

■ Fonctions

1) Commenter le résultat suivant.

```
Re[{1, 1 + I, 2 + 3 * I, Sqrt[3] I}]
```

```
{1, 1, 2, 0}
```

Réponse :
Comme la fonction Re est listable, on récupère la liste des parties réelles des éléments constituant la première liste.

2) On considère la fonction *CtoR* bien connue définie comme suit. Commenter les résultats suivants.

```
CtoR[z_] := {Re[z], Im[z]}
```

```
CtoR[{1, 1 + I, 2 + 3 * I, Sqrt[3] I}]
```

$\{\{1, 1, 2, 0\}, \{0, 1, 3, \sqrt{3}\}\}$

Réponse :
C'est la listabilité des fonctions Re et Im qui a joué : on récupère donc une première liste des parties réelles, et une seconde liste des parties imaginaires des éléments constituant la liste de départ.

```
Map[CtoR, {1, 1 + I, 2 + 3 * I, Sqrt[3] I}]
```

$\{\{1, 0\}, \{1, 1\}, \{2, 3\}, \{0, \sqrt{3}\}\}$

Réponse :
Map permet d'appliquer la fonction CtoR à chacun des élément. On récupère donc les couples {partie réelle, partie imaginaire} de chacun d'eux.

3) Donner la commande à valider afin d'obtenir le résultat suivant

```
Attributes[CtoR] = {Listable};
```

```
CtoR[{1, 1 + I, 2 + 3 * I, Sqrt[3] I}]
```

$\{\{1, 0\}, \{1, 1\}, \{2, 3\}, \{0, \sqrt{3}\}\}$

Exercice 2 : arbres

■ Expressions

1) Tracer à la main l'arbre de décomposition de l'expression suivante.

Log(| $\frac{x+I}{x-I}$ *|-argth(x))*

```
TreeForm[Log[Abs[(x + 1) / (x - 1)]] - ArcTanh[x]]
```



2) Donner l'expression *Mathematica* associée à l'arbre suivant :

```
TreeForm[(x + Tan[a] + Tan[b]) / (1 - Tan[a^2] Tan[b])]
```



3) Quelle commande *Mathematica* vous permet de vérifier votre solution ?

Réponse :
Il s'agit de TreeFrom.

■ Listes

On considère les listes suivantes :

liste1 :

```
{{{{1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}}, {{1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}}},
 {{{2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}}, {{2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}}},
 {{{3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}}, {{3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}}}}
```

liste2 :

```
{{{2 a + b}, {2 a + 2 b}, {2 a + 3 b}, {2 a + 4 b}}, {{3 a + b}, {3 a + 2 b}, {3 a + 3 b}, {3 a + 4 b}},
 {{4 a + b}, {4 a + 2 b}, {4 a + 3 b}, {4 a + 4 b}}, {{5 a + b}, {5 a + 2 b}, {5 a + 3 b}, {5 a + 4 b}}}
```

4) Quelle est la profondeur de *liste1* ? Complèter le tableau ci-dessous. En déduire pour *liste2* une commande renvoyant l'élément *3a+4b*.

Niveau	Nombres d'éléments
1	3
2	2
3	5
4	2
...	

Réponses :
Pour atteindre l'élément le plus profond, on traverse 4 accolades. La liste *liste1* est donc de profondeur 4. Pour remplir le tableau il suffit de compter les éléments contenus dans chaque groupe d'accolade (établir la décomposition en arbre sur votre brouillon pouvait aider).

On utilise cette méthode pour *liste2* (la profondeur correspondant au nombre d'indices, le nombres d'éléments permettant de se repérer).


```
liste2[[2, 4, 1]]
```

```
3 a + 4 b
```

5) Indiquer la commande utilisée pour obtenir le résultat suivant à partir de *liste1*.

```
Flatten[liste1, 2]
```

```
{{1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1}, {1, 1},
 {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2}, {2, 2},
 {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}, {3, 3}}
```

6) Donner les commandes *Mathematica* permettant de générer *liste1* et *liste2*.

Réponse :
On utilise les renseignements précédents pour trouver la structure de la liste et donc le nombre d'indices. Le nombres d'éléments pour chaque niveau indique la variation des indices. On ajuste la formule de l'expression afin qu'elle soit cohérente avec les indices.

```
liste1 = Table[i, {i, 1, 3}, {j, 1, 2}, {k, 1, 5}, {jj, 1, 2}]
```

```
liste2 = Table[a * i + b * j, {i, 2, 5}, {j, 1, 4}, {k, 1, 1}]
```

Exercice 3 : approximation

■ Représentations de rationnels et d'irrationnels

1) Donner la forme adoptée par *Mathematica* pour représenter un rationnel (on donnera un exemple).

Réponse :
Mathematica représente un rationnel sous la forme Rationnal[p,q] où p et q sont des entiers relatifs premiers entre eux.

```
FullForm[-10 / 6]
```

```
Rational[-5, 3]
```

2) Expliquer brièvement pourquoi la représentation informatique d'un nombre rationnel est très simple.

Réponse :
Il suffit de se donner un couple d'entier.

3) Quelle différence *Mathematica* fait-il entre les expressions *Sqrt[2]* et *Sqrt[2.]* ?

Réponse :
La première est en précision infinie, la seconde en précision finie (approximation d'un réel irrationnel).

4) Expliquer brièvement pourquoi la représentation informatique d'un nombre irrationnel pose problème.

Réponse :
Un nombre irrationnel possède un développement décimal illimité non périodique. On ne peut donc pas se le donner par son développement décimal : une vie ne suffirait pas à l'écrire.

■ Approximation d'un irrationnel

On considère la suite $(u_n)_{\mathbb{N}}$ suivante définie par récurrence :

$$u_1 = 2$$
$$u_{n+1} = \frac{1}{2}\left(u_n + \frac{2}{u_n}\right)$$

5) Créer une fonction f de $\mathbb{R}$ dans $\mathbb{R}$ et qui vérifie la propriété suivante :

$$u_{n+1} = f(u_n)$$

```
f[x_] := 1 / 2 (x + 2 / x)
```

6) En utilisant *NestList* établir une liste des 6 premières valeurs de $(u_n)_{\mathbb{N}}$. On nomme *val6* cette liste.

```
val6 = NestList[f, 2, 5];
```

7) En étudiant la commande et le résultat associé reproduits ci-dessous, quelle conjecture pouvez-vous faire ?

```
N[val6]
```

```
{2., 1.5, 1.41667, 1.41422, 1.41421, 1.41421}
```

Réponse :
La suite $(u_n)_{\mathbb{N}}$ converge vers $\sqrt{2}$.

8) Donner une expression *Mathematica* qui permette d'obtenir la liste de valeurs *lisval* ci-dessous.

```
lisval = Abs[(li = Table[u[n], {n, 1, 6}]) - RotateLeft[li]]
```

```
{Abs[u[1] - u[2]], Abs[u[2] - u[3]], Abs[u[3] - u[4]],
 Abs[u[4] - u[5]], Abs[u[5] - u[6]], Abs[-u[1] + u[6]]}
```

Réponse :

9) Ne connaissant pas la valeur de la limite, on cherche à connaitre le nombre de décimales exactes figurant dans u_4.
Expliquer en quoi la commande précédente permet de répondre à cette question.

Réponse :
Si la suite converge vers une limite, alors la suite définie par la suite des termes consécutifs converge vers 0.
Si l'on écrit le développement en base 10 de chacun des termes on s'aperçoit que le nombre de décimales exacte est donné par "l'exposant de la puissance de 10 moins 1" figurant dans la suite des différences successives.
Il suffit ici de regarder le 4ème élément de v*al6*.

Combien de décimales exactes a-t-on dans u_4 (on pourra s'aider de la ligne ci-dessous représentant *lisval* dans laquelle on a substitué les valeurs de u[i] à leur expression) ?

```
Abs[(val6) - RotateLeft[val6]] // N
```

```
{0.5, 0.0833333, 0.00245098, 2.1239×10^-6, 1.59486×10^-12, 0.585786}
```


Réponse :
On a 6-1=5 décimales exactes.

10) En remarquant que la suite ne prend que des valeurs rationnelles, expliquer en quoi ce qui précède apporte une réponse au problème soulevé en 4).

Réponse :
Comme chacun des éléments de la suite est un nombre rationnel, il est facilement codable (*Cf* question 2.). Pour connaitre ici $\sqrt{2}$ à une précision donnée, il suffit de calculer la suite jusqu'à un certain rang. On connait donc "toutes" les valeurs approchées de $\sqrt{2}$ ...mais toujours pas $\sqrt{2}$ en précision infinie.

Exercice 4 : géométrie affine complexe

On considère la similitude de centre *O*, d'angle *theta* et de rapport *k* suivante :

$$sim : \mathbb{R}^2 \longrightarrow \mathbb{R}^2$$
$$\begin{pmatrix} x \\ y \end{pmatrix} \longmapsto k \begin{pmatrix} \cos(\text{theta}) & -\sin(\text{theta}) \\ \sin(\text{theta}) & \cos(\text{theta}) \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix}$$

■ Dans le plan complexe

1) Définir la fonction *Mathematica* complexe associée

```
sim[z_] := k * z * Exp[I * theta]
```

2) Créer une liste *pointsAlea* de 5 points complexes au hasard

```
pointsAlea = Table[Random[Complex], {5}];
```

Dans la suite on pose k=3/4 et theta=π/8

3) En déduire une liste *pointsSim* contenant l'image des points complexes précédents sous l'action de la similitude

```
k = 3 / 4; theta = Pi / 8;
```

```
pointsSim = Map[sim, pointsAlea];
```

4) Créer une liste *pointsIt* contenant les images de *pointsAlea* sous l'action des composées de la similitude jusqu'à l'ordre *12*.

```
pointsIt = NestList[sim, pointsAlea, 12];
```

■ Dans le plan réel

5) Donner la commande permettant d'afficher la figure composée en reliant par des segments, et dans l'ordre, les points de *pointsAlea*.

```
CtoR[z_] := {Re[z], Im[z]}
Attributes[CtoR] = {Listable};
```

```
Show[Graphics[Line[CtoR[pointsAlea]]]];
```



6) Donner la commande permettant l'affichage des treize figures images de la précédente (*indication :* elles ont été obtenues à la question 4 et sont contenues dans la variable *pointsIt*).

```
Show[Graphics[Map[Line, CtoR[pointsIt]]], AspectRatio → Automatic];
```



Exercice 5 : booléens

On considère dans cette partie les quatres cercles *C1, C2, C3* et *C4* de centre *(±1,±1)* et de rayon $\sqrt{3}$ suivants.

```
<< Graphics`ImplicitPlot`
```



```
grcer = ImplicitPlot[{(x - 1)^2 + (y - 1)^2 - 3 == 0, (x + 1)^2 + (y - 1)^2 - 3 == 0,
    (x + 1)^2 + (y + 1)^2 - 3 == 0, (x - 1)^2 + (y + 1)^2 - 3 == 0}, {x, -3, 3},
   PlotStyle → {Thickness[.02], Thickness[.012], Thickness[.01], Thickness[.001]},
   DisplayFunction → Identity];
grLeg = Graphics[{Map[Point, {{1, 1}, {-1, 1}, {1, -1}, {-1, -1}}],
    Text["C1", {2.5, 2.5}], Text["C2", {-2.5, 2.5}],
    Text["C3", {-2.5, -2.5}], Text["C4", {2.5, -2.5}]}];
Show[{grcer, grLeg}, DisplayFunction → $DisplayFunction];
```



1) Donner l'équation cartésienne implicite de *C1*.

Réponse :
C1 est le cercle de rayon $\sqrt{3}$ et de centre *(1,1)*. Son équation cartésienne implicite est donc $(x-1)^2+(y-1)^2=3$.

2) Créer une fonction booléenne *f1* qui renvoie *True* si un point *{x,y}* est dans l'intérieur du cercle *C1, False* sinon.

```
f1[{x_, y_}] := ((x - 1)^2 + (y - 1)^2 - 3 < 0)
```

3) Créer une liste *tapis* contenant *5000* coordonnées aléatoires de points comprises entre *-1* et *1*.

```
tapis = Table[Random[Real, {-1, 1}], {5000}, {2}];
```

On considère données des fonctions *f2, f3* et *f4* similaires à *f1* mais pour les cercles *C2*, *C3* et *C4* respectivement.

4) Utilisant les fonctions booléennes élémentaires et les fonctions *f1, f2, f3* et *f4,* construire une fonction *FF* qui renvoie *True* si un point *{x,y}* est dans la région grisée suivante.

```
f2[{x_, y_}] := ((x + 1) ^2 + (y - 1) ^2 - 3 < 0)
f3[{x_, y_}] := ((x + 1) ^2 + (y + 1) ^2 - 3 < 0)
f4[{x_, y_}] := ((x - 1) ^2 + (y + 1) ^2 - 3 < 0)
```




```
FF[{x_, y_}] := Or[And[f1[{x, y}], f3[{x, y}]], And[f2[{x, y}], f4[{x, y}]]]
```

5) Utilisant ce qui précède, créer une liste *ptSel* qui contienne les points de *tapis* contenus dans la région grisée.

```
Show[{pts, grcer, grLeg}];

ptSel = Select[tapis, FF];
```

6) Afficher ces points.

```
pts = Show[Graphics[Map[Point, ptSel]], AspectRatio → Automatic];
```